# Contents

1 Introduction					
	1.1	Functional programming	4		
	1.2	Downloading and building Angort	5		
2		ting started: immediate mode	6		
	2.1	Immediate mode	6		
	2.2	Reverse Polish notation and the stack	6		
	2.3	Running scripts	6		
	2.4	Words and functions	7		
	2.5	Types and literals	7		
		2.5.1 Coercions	9		
	2.6	Defining new functions	9		
		2.6.1 Word parameters and local variables	9		
		2.6.2 Word documentation strings and stack pictures	10		
3	Basi	ic built-in words	11		
	3.1	Stack manipulation	11		
	3.2	•	11		
	3.3		12		
			12		
4	Flov	v control	12		
	4.1		 12		
	4.2		13		
	4.3		14		
	4.4		14		
	1.1	•	14		
			15		
			16		
	4.5	1	17		
	4.6		17		
	4.7		17		
	4./	Explicit iterators	1/		
5	Glo	bal variables	18		
6	Con	stants	19		
7	The	true nature of function definitions	19		
	7.1	Sealed functions	19		
	7.2		19		
	7.3	•	20		
8	List	s	20		
			_		
9	Sym	abols	21		
10	Has		21		
			22		
			22		
	10.3	Hash to string function	23		
11	L Garbage collection				

12 Functional programming				
	12.1 Recursion	24		
	12.1.1 A warning	24		
	12.2 Words for dealing with functions			
	12.3 Closures	25		
	12.3.1 Closures are by reference	25		
	12.3.2 Objects with delegates and private members using hashes and closures	26		
13	Exceptions	26		
14	More built-in words	27		
15	Getting help	27		
16	Namespaces and packages	27		
	16.1 Importing packages into the default space	27		
	16.1.1 Local packages	28		
	16.2 Namespaces, libraries and modules	28		
<b>17</b>	Plugin libraries (currently Linux only)	28		
18	Topics missing from this document	29		

### 1 Introduction

Angort<sup>1</sup> is a stack-based concatenative programming language, based on the venerable Forth, with some extra features. These include:

- variables and function parameters
- higher order functions with (nearly) full lexical closure;
- lists and hashes (dictionaries) with garbage collection;
- native C++ plugin architecture.

Here are some examples. First, the familiar recursive quicksort algorithm:

```
# Colon at the start of a line introduces a named function definition.
# This function has one parameter "lst" and one local variable "piv".
:qs |lst:piv|
    # if length of list is <=1...
    ?lst len 1 <= if
        # return the list itself
        ?lst
   else
        # otherwise remove the first item as the pivot
        ?lst pop !piv
        # quicksort the list of items less than the pivot
        ?lst (?piv <) filter qs
        # add the pivot to that list
        [?piv] +
        # quicksort the list of items greater than the pivot
        # and add it.
        ?lst (?piv >=) filter qs +
        # resulting list is left on the stack, and so returned.
   then;
```

<sup>&</sup>lt;sup>1</sup>The name is an entirely random pair of syllables, it has no significance.

Next, a program to read a CSV file and print the sums of all the columns:

```
# load the CSV plugin but do not import the symbols into the
# default namespace, so we have to access them with csv$...
'csv library drop
# load the CSV file - creates a CSV reader object which will read
# into a list of hashes where all columns are floats, and uses it
# to read the file into the global "CSV".
[% 'types "f" ] csv$make "magic.log" csv$read !CSV
# make a list of keys from the first item in CSV.
[] ?CSV fst each {i,} !Keys
# create a "slug" (an anonymous function which runs immediately,
# to provide local variables and multiline flow control)
# with a local variable i.
(|:i|
    # for each key, store the iterator value in "i" so it's
    # accessible within a closure
    ?Keys each { i!i
        # create a pair consisting of the iterator (i.e. key name)
         # and the sum of the key's values, done by using map/reduce:
         # the map extracts the key's values, the reduce performs the sum.
         0 ?CSV (?i swap get) map (+) reduce
        # format the list and print it.
        "%s %f" format.
)@ quit # run the slug and quit
```

Angort combines the power and ease of a modern dynamic language with the convenience of a Forth-like language, and has been used in various applications including:

- building command/control/monitoring environments for robotic systems;
- scripting experiment runs for neural networks;
- writing fairly complex data analysis tools;
- generating visualisations of neural networks and other data;
- algorithmically generated music;
- simple 2D games.

Because the interpreter is interactive and upon completion of a script drops back to a command interface, it is very useful for building domain-specific control languages. For example, our ExoMars locomotion prototype is controlled with Angort, and its script includes the following definitions:

```
# define a constant "wheels" holding a range from 1-6 inclusive
range 1 7 const wheels
# define a new function "d" with a single parameter "speed"
:d |speed:|
    # set a help text for this function
    :"(speed --) set speed of all drive motors"
    # for each wheel, set the required speed to the value
    # of the parameter
   wheels each {
       ?speed i!drive
# slightly more complex function for steering
:t |angle:|
   :"(angle --) turn front wheels one way, back wheels opposite way"
   ?angle dup 1!steer 2!steer
    0 0 3!steer 4!steer
    ?angle neg dup 5!steer 6!steer
# define a function to stop the rover by setting all speeds to zero
```

Once these words are defined we can steer the robot in real time with commands like:

```
2500 d
30 t
```

These will set the rover speed to 2500, turn it to 30 degrees, and stop it respectively. We can also directly type things like:

```
wheels each { i dactual .}
```

which will print the actual speeds of all the wheels.

### 1.1 Functional programming

It's also possible to perform some functional programming with anonymous functions:

```
:sum |list:| 0 ?list (+) reduce;
```

will allow us to sum a list and print the results:

```
[1,2,3,4,5] sum .
or even:
1 1001 range (dup*) map sum .
```

to print the sum of the squares of the first 1000 integers. As can be seen, Angort is a very terse language.

In the examples given so far, functions such as dactual, !drive and ?drive are links to native C++ code: it is very easy to interface Angort with C++.

### 1.2 Downloading and building Angort

Angort can be downloaded from <a href="https://github.com/jimfinnis/angort">https://github.com/jimfinnis/angort</a>. Once downloaded, it can be built and installed with the following commands (from inside the top-level Angort directory):

```
mkdir build
cd build
cmake .. -DCMAKE_BUILD_TYPE=Release
make
sudo make install
```

This is for a Linux machine with CMake and the readline development libraries. The interpreter will then be installed, typically as /usr/local/bin/angort. A small set of Angort libraries (i.e. libraries written in Angort) will also be installed into /usr/local/share/angort.

A set of native C++ plugin libraries can also be downloaded from https://github.com/jimfinnis/angortplugins. Once Angort has been installed, these can be built and installed with

```
./buildall sudo ./install
```

They will also be installed into /usr/local/share/angort. Using reallybuildall will attempt to build extra libraries which require additional packages, such a a CURL interface, SDL graphics support and JACK MIDI support. Many of these have been written for somewhat esoteric purposes as I have needed them.

# 2 Getting started: immediate mode

This section will describe the basic concepts behind the language, such as reverse Polish notation and the stack, and introduce using the language in immediate mode.

#### 2.1 Immediate mode

In immediate mode, each line of Angort is compiled and run straight away. This is how the Angort executable starts when it is not given an Angort script's filename on the command line. Angort will also drop back to this mode if a script completes without quit being called. Running the interpreter without a command line will give a prompt:

```
1 | 0 >
```

The two numbers are the number of garbage-collectable objects in the system and the number of items on the stack, respectively. The interpreter is in "immediate mode", as opposed to "compilation mode" — any text entered will be compiled to bytecode and run when enter is pressed, rather than being added to a function definition.

#### 2.2 Reverse Polish notation and the stack

Angort is a stack-based language: there is a single stack containing values, and most functions change the contents of the stack in some way. For example,

3

by itself will just put the value 3 on the stack. Then

.

will pop the value from the top of the stack and print it.

```
3 4 +
```

will push 3 and 4 onto the stack, then add them together replacing them with 7, and then print the 7. This kind of notation is often referred to as **reverse Polish notation** (RPN) as opposed to the more common **infix** notation. More complex expressions are built out of sequences of operations on the stack. For example, the expression

$$\sin(5 + 32 + \sqrt{43 \times 12})$$

would be written as

```
43\ 12 * sqrt 32 + 5 + sin
```

This is a little difficult at first, but rapidly becomes second nature<sup>2</sup>

### 2.3 Running scripts

Scripts are traditionally<sup>3</sup> given the extension .ang, but this appears nowhere in the codebase. To run a script, create a file of Angort function definitions and commands and pass it as an argument to the angort command. Each line will be parsed and run in order, unless it is a function definition in which case the compilation will occur at the end of the definition.

Note that each line of Angort (outside a function) is parsed and run immediately. This means that multiline flow control constructs (such as loops and if..then) will not work unless they are inside a function. This is easily solved by using anonymous functions, as described in Section 4.2.

<sup>&</sup>lt;sup>2</sup>Most old calculators work using RPN, and quite a few of the more powerful programmables still do.

<sup>&</sup>lt;sup>3</sup>A somewhat unusual word to use when there is currently only one user of the language.

At the end of a script, control will drop into immediate mode. This is often useful when Angort is used to provide a domain-specific control language: the script sets up a large number of commands and perhaps initialises native C++ libraries, and then waits for the user's commands. If this is not required, terminate the script with the quit command. This will cause Angort to exit with the return code 0.

#### 2.4 Words and functions

Most of the Angort language consists of words which each perform a single, isolated task – there is very little syntax. Even common "syntactic" keywords like if (for conditions), [] (for creating lists) and + (for addition) operate primarily through stack operations, although if does some more complex things at compile time (such as calculating jump offsets). Code in Angort largely consists of many of these words chained together to perform a task. These words can be combined into user-defined functions.

In general, I will use "word" to refer to a single Angort built-in token, particularly if it compiles to a single opcode; while "function" will refer to user-defined functions. However, I may vary – there is no real distinction, since an Angort function call is itself a single opcode.

### 2.5 Types and literals

Angort is a dynamically typed language, with type coercion (weak typing). The following types are available:

Name	Definition	Example of literal
None	The nil object, specifying "no value"	none
Integer	32 or 64-bit integers depending on architec-	5045
	ture	
Long	depends on architecture	50451,32L,0ffx, 45ao,
		10111b, 3200ffffhl
Float	32-bit floats	54.0
Double	64-bit floats	54.01,32.2L
String	Strings of characters	"Hello there"
Symbol	Single-word strings, internally stored as in-	`foo
	tegers and generally used as hash keys	
Code	Blocks of Angort code, anonymous func-	( dup * 1 + )
	tions	
Integer range	A range of integers between two values with	0 4 range <sup>1</sup>
	optional step	
Float range	A range of floats between two values with	0 4 0.1 frange
	step	
List	A array/list of values <sup>2</sup>	[1,2,"foo"]
Hash	A map of values to values implemented as a	[% `foo 1, "fish"
	hash table, where keys can strings, symbols,	"fish"]
	integers or floats	

"Code" above actually conflates two types internally — codeblocks, which have no environment; and closures, which consist of a codeblock and some stored variables. They appear identical to the user. There are some other types used internally, such as the types for iterators and deleted objects in a hash.

Integers can also have a base character, 'b', 'x'/'h', 'o' or 'd', which goes before the optional 'I' for 'long.' **Note that** all numbers must start with a digit, so a leading zero will be required for some hex numbers (e.g. '0ffh' not 'ffh').

<sup>&</sup>lt;sup>1</sup>This is actually a call to the range function which takes two integers to create a range.

<sup>&</sup>lt;sup>2</sup>Stored internally as an automatically resized array rather than a linked list.

Checking and getting types (isnone, isnumber, iscallable, type) heredocs

#### 2.5.1 Coercions

- Integers and floats are coerced to strings in string contexts.
- In binary operations, if one of the operands is a float, both are coerced to floats:

```
1|0> 1 2 / .
0
1|0> 1.0 2 / .
0.500000
```

- In certain binary operations (currently just "+") if one of the operands is a string, both will be coerced to strings.
- more... longs and doubles, notably, and what happens when you can't

### 2.6 Defining new functions

New functions are defined with code of the form

```
:functionname ...;
A simple example is:
```

```
:square dup *;
```

This will define a function square which will duplicate the value on top of the stack, multiply the top two values (effectively squaring the number<sup>4</sup>) and exit, leaving the squared number on the stack. This can then be used thus:

```
3 square 4 square + .
```

which will print 25, i.e.  $3^2 + 4^2$ . Note that leaving values behind on the stack on exiting a function is how we return values in Angort. While defining a function, Angort is is compilation mode — functions will be converted to bytecode but added to the definition of the new function rather than executed immediately. Function definitions can therefore span more than one line:

This example uses recursion and the Angort if..else..then construction. This looks a little odd to programmers not familiar with Forth – please see Section 4.1 for more details. Mention how immediate and script mode work with definitions. Add parameter typing.

#### 2.6.1 Word parameters and local variables

Until now, all the code has been perfectly valid Forth<sup>5</sup>. However, the manipulation of stack required in Forth is a challenge (and modern computers have a little more memory), so Angort has a system of named parameters and local variables. These can be defined by putting a special block of the form

```
|param1,param2,param3...: local1,local2,local3|
```

<sup>&</sup>lt;sup>4</sup>This will produce an error if the value is not a number.

<sup>&</sup>lt;sup>5</sup>With the exception of the factorial function, which uses local variables and recursion.

after the function name in the definition. Locals and parameters are exactly the same internally, but the values of parameters are popped off the stack when the new function is called.

Once defined, locals (and parameters) can be read (i.e. pushed onto the stack) using a question mark followed by the variable name. Similarly, a local is written (i.e. a value popped off the stack and stored in the local) by using an exclamation mark. For example,

will read the two arguments into the locals x and y, add them, store the result in the local z, and then exit, throwing everything away. Note that a function with parameters but no locals is defined by leaving the part after the colon empty:

```
:magnitude |x,y:|
    ?x dup *
    ?y dup * +
    sqrt
;
```

while leaving the part before the colon empty will define a function with locals but no parameters:

Here, the curly brackets enclose a loop, which is exited by the ifleave instruction when the stack top value is 10. See Section 4.4 for more details. Types in param lists. Returning values.

#### 2.6.2 Word documentation strings and stack pictures

Following the locals and parameters (or just the function name if there are none) there may be a function documentation string. It has the form

```
:"(before -- after) what the function does"
```

The section in brackets is known as a *stack picture*, <sup>6</sup> and describes the action of the function on the stack. The part before the hyphen shows the part of the stack removed by the function, and the part after shows its replacement. For example:

```
| :dist |x1,y1,x2,y2:|

:"(x1 y1 x2 y2 -- distance) calculate the distance between two points"

?x1 ?x2 - abs dup * # this is (x1-x2)^2

?y1 ?y2 - abs dup * # this is (y1-y2)^2

+ sqrt # sum them, and find the root
```

Note that the names in the picture are not necessarily those of variables: they are single word descriptions of values. Typically these will be variable names for the "before" part, but not for the "after" part (since the return values are simply what's left on the stack).

To print the documentation for a word, enter ??wordname at the prompt. This will also work for most built-in Angort functions, but not all<sup>7</sup>. It will also work for any functions inside C++ plugin libraries (see Section 17). For example:

<sup>&</sup>lt;sup>6</sup>Or sometimes a *stackflow symbol*.

<sup>&</sup>lt;sup>7</sup>This is because most Angort functions compile to calls to C++ functions, which can have descriptions. Simpler operations, such as if..then..else, loop markers, arithmetic and the "." print operator, compile directly to bytecode opcodes. These cannot have descriptions.

```
Angort version 2.9.2 (c) Jim Finnis 2012-2017
Use '??word' to get help on a word.

1|0 > ??redir
redir: (filename --) open a new file and redirect output to it.
For more complex file output, use the IO library. On fail, throws ex$failed.

1|0 >
```

### 3 Basic built-in words

### 3.1 Stack manipulation

There exist a number of words whose sole purpose is to manipulate the stack. These words, and their stack pictures, are given in the table below.

name	stack picture
dup	(a-a a)
drop	(a –)
swap	(a b - b a)
over	(a b – a b a)

Note that this is a much smaller set than the set of Forth stack words, which includes rot, roll, nip and tuck. The local variable system makes such complex operations unnecessary.

### 3.2 Binary operators

In the following operations, these conversions take place:

- If one of the operands is a string, the other will be converted to a string. Only "+", "\*" and comparison operators will be valid.
- If one of the operands is a long and the other is an integer, the integer will be converted to a long and the result will be a long.
- If one of the operands is a float and the other is an integer or long, that will be converted to a float and the result will be a float.
- If one of the operands is a double and the other is an integer, long, or float, that will will be converted to a double and the result will be a double.
- The comparison operators will do identity checks on objects (lists, ranges etc.), not deep comparisons.
- The comparison functions actually return integers, with non-zero indicating falsehood.

name	stack picture	side-effects and notes
+	(a b – a+b)	
-	(a b - a-b)	
*	(a b - a*b)	ints are coerced to floats if one operand is an int, but not otherwise.
		"string int *" produces a repeated string, but not "int string *".
/	(ab-a/b)	
%	(a b - a%b)	integer remainder ("mod") operator
>	(a b - a > b)	string comparison works as expected
<	(a b - a < b)	string comparison works as expected
=	(a b - a=b)	equality test: string comparison works as expected
!=	(a b - a!=b)	inequality test: string comparison works as expected
and	(a b − a∧b)	binary and – inputs are coerced to integers, nonzero is true
or	(a b − a∨b)	binary or – inputs are coerced to integers, nonzero is true
band	(a b – a & b)	bitwise AND
bor	(a b – a   b)	bitwise OR
bxor	$(a b - a \sim b)$	bitwise XOR

### 3.3 Unary operators

name	stack picture	side-effects and notes	
not	(a – !a)	logical negation of a boolean	
neg	(a – -a)	arithmetic negation of float or integer	
abs	(a -  a )	absolute value	
isnone	(a – bool)	true if value is none	
bnot	(a – !b)	bitwise not	

#### 3.3.1 Input and output

Increase the meaning!

- dot, p, nl
- errp, errnl
- redir, endredir
- rawp, x
- format

### 4 Flow control

This section of the manual will describe how Angort handles conditional code and loops. There are also several idioms using anonymous functions which can be used for flow control, which will be described briefly.

### 4.1 Conditions


These have the form:


```
<condition> if <runs if true> then
```

The word if pops the item off the top of the stack, and jumps forward to just after the corresponding then if it is false (zero). There is also a two way branch:

```
<condition> if <runs if true> else <runs if false> then
```

This works the same way, except that the if jumps to after the else if the top value is false, and else jumps forward to just after then. This is shown in the figure below:





Here is a code example, printing whether a number is even or odd:

We can run this with various values:

```
1|0> 4 evenodd
number is even
1|0> 3 evenodd
number is odd
```

Conditions can be nested.

### 4.2 Multi-line flow control and "slugs"

It is important to note that Angort processes input on a line-by-line basis when not defining a function, both within a script and in immediate mode. Therefore, code like the following will not work:

```
read len 0 = if
"zero length string"
then
```

while the following will:

```
read len 0 = if "zero length string" then
```

With the former code, Angort will attempt to compile and run the first line, and will find that the line has no matching then. This will not occur on the latter example because the then is provided on the same line. This will not happen while a function is being defined, because Angort suspends compilation until the function is completed with a semicolon.

While this is a limitation when writing scripts it is easily surmounted by either putting code into a named function, or defining and immediately running an anonymous function. Anonymous functions are enclosed with parentheses, leaving the function on the stack. They can then be run with call or @ (the two are equivalent). Thus the above can be written

```
read len 0 = if
    "zero length string"
then

0 e
```

In informal conversations I have taken to referring to anonymous functions which are called immediately as "slugs," both from the typesetting terminology and from the visual appearance of  $(\ldots)$  @.

#### 4.3 Case blocks

In order to avoid testing if..else..then constructions where a number of different, mutually exclusive cases need to be tested, the "case block" structure is provided, which emulates the "elsif" construction in other languages.

Case blocks begin with cases, and each case consists of

```
<condition> if <action> case
```

The case block must end, after the final case, with

```
<default action> otherwise
```

An example:

```
cases
    ?a 0 = if "It's Zero". case
    ?a 1 = if "It's One". case
    ?a 2 = if "It's Two". case
    ?a 3 = if "It's Three". case
    ?a 4 = if "It's Four". case
    ?a 5 = if "It's Five". case
    ?a 6 = if "It's Six". case
    ?a 7 = if "It's Seven". case
    "It's something else". otherwise
;
```

This function will print an appropriate message for integers from 0 to 7, and "It's something else" for any other value.

#### 4.4 Loops

There are two kinds of loops in Angort — infinite loops, which must be broken out of using <code>leave</code> or <code>ifleave</code>; and iterator loops, which loop over a collection (i.e. a hash or list) or a range. Both are delimited with curly brackets {}, but iterator loops use the <code>each</code> word before the opening brace.

#### 4.4.1 Infinite loops

Any leave word will break out of an infinite loop:

```
if leave then # if so, leave the loop
}
```

This will count from 0 to 99. Note the use of an anonymous function as described in Section 4.2, which contains a local variable counter. The sequence

```
if leave then
```

is so common that it is coded to a special word of its own, and can be written as

```
ifleave
```

The function above could be written tersely as

```
| (|:c| 0!c {?c. ?c 1+ !c ?c 100= ifleave)@;
```

This is still using a slug because we require the local variable provided by the anonymous function. Loops can be nested, and the leave words will jump out of the innermost loop in which they appear.

#### 4.4.2 Iterator loops

Iterator loops loop over an *iterable* value. These are currently ranges, hashes and lists. They are delimited by curly braces like infinite loops, but are preceded by the each word. This pops the iterable off the stack and makes an iterator out of it, over which the loop runs. The iterator exists for as long as the loop does, and is destroyed when the loop completes. Again, iterator loops can be nested (and can be nested with infinite loops). If we use an iterator loop to iterate over a range object, the counting example can be rewritten as:

```
( 0 100 range # stack a range from 0-99 (see below)
each { # start the iterator loop
i # push the current item in the iterator
. # print it
} # end of loop
]0
```

Here, the range word takes two integers a,b from the stack which define an integer range  $\{x \in \mathbb{Z} \mid a \leq x < b\}$ . Note that range end value is exclusive — that is, it is the first value which is *not* in the range — so the above example, specified as 0 100 range, actually counts from 0 to 99. This looks somewhat odd with integer ranges, but serves to prevent float ranges relying on equality tests:

```
( 0 10 0.09 frange # float range from 0-10 in steps of 0.09 each {i.} # print them
```

will print about 9.990008 as its final value. Because this doesn't require any local variables, it works as an immediate mode one-liner:

```
0 10 0.09 frange {each i.}
```

Although we've not covered lists and hashes yet, it's useful to know that we can iterate over the values in a list. For example

```
[1,2,3,4,5] each {i.}
```

will print the numbers from 1 to 5. Similarly, we can iterate over the keys and values of a hash:

```
|| [% "foo" "bar", "baz" "qux"] each \{i.\} will print the hash's keys: "foo" and "bar", while
```

```
|| [% "foo" "bar", "baz" "qux"] each {i p "=" p ival.}
```

will print "key=value" for each entry. Here, ival gets the value (as opposed to the key) of the current pair in the iterator, and p prints without a newline.

#### 4.4.3 Nested iterator loops

Each loop has its own iterator, even if they come from the same iterable:

The range here is used twice, generating two separate iterators with their own current value.

While the word i retrieves the iterator value inside the current loop, inside nested iterator loops it is often useful to get the value in outer loops. This can be done with the j and k words, to get the outer and next outer iterator values:

The jval and kval words also exist to get a hash iterator value for two outer iterator loops.

### 4.5 The state of the stack inside a loop

Note that the iterator is not kept on the normal stack — it is popped off when the loop starts and kept inside a special stack. The same applies to infinite loops: no loop state is kept on the stack. This leaves the stack free for manipulation:

### 4.6 Words to create ranges

There are three words to create a range on the stack:

name	stack picture	side-effects and notes	
range	(x y – range)	create an integer range over the interval $[x, y)$ with a step of 1; i.e.	
		a range from $x$ to $y-1$ inclusive.	
srange	(x y s – range)	create an integer range over the interval $[x, y)$ with a step of $s$	
frange	(x y s – range)	create an float range over the interval $[x, y)$ with a step of $s$	

See Section 2.6.2 for how stack pictures define the parameters and return values for a function.

### 4.7 Explicit iterators

It's sometimes necessary to use iterators manually instead of creating them automatically with "each" and getting the values using "i" in an iterator loop. To do this, we have the following words:

name stack picture		side-effects and notes	
mkiter	(iterable – iterator)	create an iterator	
icur	(iterable – item)	get the current item	
inext	(iterable –)	advance the iterator	
idone	(iterable – bool)	return 1 if the iterator is at the last item	
ifirst	(iterable –)	reset the iterator to the start	
iter	(– iterable)	return the iterator which is being iterated over in the current loop	

You might need this when you want to iterate over two collections in parallel, for example in implementing the zipWith function<sup>8</sup>. This function takes two collections, and runs a binary function on pairs of items, each from one of the collections, returning a list:

```
["foo","bar"] ["fish","zing"] (+) zipWith each{i.}
would print
    foofish
    barzing
```

This could be implemented by using an index to go over both lists:

<sup>&</sup>lt;sup>8</sup>zipWith is already built into Angort, this is just an illustrative example.

```
:zipWith |a,b,f:|
    []
    0 ?a len ?b len min range each {
        i ?a get
        i ?b get ?f call ,
    }
;
```

but a more elegant solution might be:

```
:zipWith |a,b,f:it|
    ?b mkiter !it
    []
    ?a each {
        ?it idone ifleave
        i ?it icur ?f@,
        ?it inext
    }
;
```

Here, we use an each loop to iterate over list a, and an explicit iterator to iterate over list b.

### 5 Global variables

Global variables are defined in two ways. The "polite" way is to use the global keyword, which creates a new global of the name following it, initially holding the nil value none:

```
1|0> global foo
1|0> ?foo.
NONE
1|0> 5!foo
1|0> ?foo.
```

The other way to define globals is simply to access a variable whose name begins with a capital letter. If no global or local exists with that name, a global is created with the initial none value:

```
1|0> 5 !Foo
1|0> ?Foo.
```

Globals, unlike locals, do not require a ? sigil to be read: if the name is used "bare", their value will be stacked just as if the name was preceded with ?, unless they contain an anonymous function. In this case, the function is run (see Section 7 below). It is, however, good practice to use the sigil. Writing to a global is done using the familiar! sigil.

Finally, accessing a global using its name without a sigil will do nothing if it contains None, while accessing it with the ? will stack the None:

```
1|0 > 10!A
1|0 > ?A.
10
1|0 > A.
10
1|0 > A.
10
1|0 > A.
from [dot] <stdin>:5/2
Error: stack underflow in stack 'main'
Last line input: A.
1|0 > ?A.
NONE
1|0 >
```

#### 6 Constants

Constants are similar to globals, but with the following differences:

- they are defined and set using the const keyword this will pop a value off the stack and set the new constant to that value;
- they can never be redefined or written to.

Here are some examples which might be found at the start of a maths package:

```
3.1415927 const pi
2.7182818 const e

180 pi/ const radsToDegsRatio
pi 180/ const degsToRadsRatio

:degs2rads
    :"(degs -- rads) convert degrees to radians"
    degsToRadsRatio*
;

:rads2degs
    :"(rads -- degs) convert radians to degrees"
    radsToDegsRatio*
;
```

### 7 The true nature of function definitions

Words are actually global variables bound to anonymous functions. Given that anonymous functions are written as blocks of Angort in brackets (see below), then

```
||:square |x:| ?x ?x *;

could also be written as

|| global square | (|x:| ?x ?x *) !square
```

with exactly the same end result. Referring to a global by using the ? sigil will simply stack its value, whereas referring to it without the sigil will check if it is holds a code block or closure and run it if so, otherwise stack the value. This is useful in functional programming.

#### 7.1 Sealed functions

Combined with constants, this allows for "sealed" function definitions. In general all functions can be redefined, but a function defined thus:

```
| (|x:| ?x ?x *) const square
```

cannot be. This can be useful in package development.

### 7.2 "import as" semantics

This is also useful when you want to import only certain names from a plugin library (Section 17). Consider the vector2d library, which contains the functions vector2d\$x and vector2d\$y. These are cumbersome, so we could import just these in their short forms as

```
?vector2d$x const x
?vector2d$y const y
```

Using the ? ensures that we get the function itself, rather than trying to run it<sup>9</sup>.

### 7.3 Forward declarations (deferred declarations)

It's also possible to use this mechanism to define global variables, use them as function names, and change their values later. This lets us defer function definitions:

```
global foo  # define a global called ''foo'' (null-valued)
:bar foo;  # a function which uses it

:foo...;  # actually define foo
```

#### 8 Lists

Angort lists are actually array based, with the array resizing automatically once certain thresholds are passed (similar to a C++ vector or Java ArrayList). A list is created by enclosing Angort expressions in square brackets and separating them by commas:

```
[]  # the empty list
[1,2,3]  # a list of three integers
["foo","bar",1] # two strings and an integer
```

As noted above, lists can be iterated over. Lists can also contain lists, and can be stored in variables — more precisely, references to lists can be stored in variables:

```
1|0> [1,2] !A  # create a list and store it in global A
2|0> ?A!B  # copy A to B
2|0> ?A push  # append an item to A
2|0> ?A each {i.}  # print A
1
2
3
1|0> ?B each {i.}  # print B - it also has the extra item!
1
2
3
```

Note that the list in B has also changed — it is the same list, just a different reference. The following are the words which act on lists, with their stack pictures:

name stack picture side-effects and notes		side-effects and notes
[ (– list)		creates a new list
,	(list item – list)	appends an item to the list
]	(list item – list)	appends an item to the list
get	(n list – item)	get the nth item from the list
set	(item n list –)	set the nth item in the list
remove	(n list – item)	remove and return the nth item
shift	shift (list – item) remove and return the first item	
unshift	(item list –)	prepend an item
pop	(list – item)	remove and return the last item
push	(item list –)	append an item
in	(item iter –)	return true if an item is in a list, integer range or hash keys

Note that the literal notation for lists — the square brackets and the comma — fall naturally out of the definition of the words<sup>10</sup>. The comma is a useful word, acting as a way to add items to a list on top of the stack without popping that list. This means we can write code to do a list copy:

<sup>&</sup>lt;sup>9</sup>In reality we would probably use the optional list parameter to import, as described in Section 16

<sup>&</sup>lt;sup>10</sup>There is an exception: if the tokeniser finds the sequence [] it discards the second bracket, allowing us to notate the empty list in a natural way.

This syntax allows us to create a form of list comprehension:

```
[[1,2,3,4]] each {i dup * dup 2 % if , else drop then}
```

will create the list [1,9]: the squares of the numbers but only if they are odd. It does this by squaring the number, duplicating and testing to see if it is nonzero modulo 2, and if so appending it to the list, otherwise dropping it. In reality, we would probably use the map and filter words with anonymous functions:

```
| [1,2,3,4] (dup *) map (2 %) filter
```

# 9 Symbols

Symbols are single-word strings which are stored in an optimised form for easy and quick comparison (they are turned into unique integers internally). They are specified by using a backtick (') in front of the word. They're most useful as keys in hashes, covered in the next section.

#### 10 Hashes

Hashes are like Python dictionaries: they allow data be stored using keys of any hashable type (integers, floats, strings, symbols – anything except hashes and lists). Hashes are created using a similar syntax to the list initialiser, but with a % after the closing brace and data in key,value pairs<sup>11</sup>:

[%]

creates an empty hash, and

```
'foo "the foo string",
    'bar "the bar string"
]
```

creates a hash with two entries, both of which are keyed by symbols (although the keys in a hash can be of different types). We can add values to the hash using set which has the picture (val key hash --):

and read values using get which has the picture (key hash -- val):

```
2|0> 'foo ?H get 2|1> . the foo string
```

We can also iterate over the hash's keys and values:

<sup>&</sup>lt;sup>11</sup>This is a somewhat awkward syntax, but all the other bracket types were used elsewhere.

If we run this on the hash H defined above, we get

```
2|0> ?H dumphash
foo: the foo string
bar: the bar string
```

kval We can nest iterator loops for hashes too, as we did with lists. Just as i, j and k get the inner, next outer and next outer loop keys (see Sec. 4.4.3), ival,jval and kval get the inner, next outer and next outer loop values with hashes.

### 10.1 Shortcut symbol get and set

There is a syntactic sugar for retrieving the value of a key in a hash where the key is a literal symbol. Instead of using

```
`foo ?H get
we can use
?H?`foo
```

This has been added because this is by far the most common use-case. We also have the same ability to set a value in a hash with a literal symbol:

```
96 ?H! 'temperature instead of 96 'temperature ?H set
```

#### 10.2 Words for hashes

Hashes can also use many of the same words as lists:

name	stack picture	side-effects and notes
[%	(– hash)	creates a new hash
,	(hash key value – hash)	adds a value to the hash
]	(hash key value – hash)	adds a value to the hash
get	(key hash – value)	get a value from the hash, or none if it is not present
set	(value key hash –)	set a value in the hash
remove	(key hash – value)	remove and return a value by key
in	(key hash –)	return true if a key is in the hash

Note that the comma and close bracket words examine the stack to determine if they are working on a list or a hash.

### 10.3 Hash to string function

By default, printing a hash — or converting it to a string any other way — will just print the default string. This tells you it's a hash and gives its address in memory:

```
1|0 > [%] . 
<TYPE hash:0x16a4840>
2|0 >
```

However, if we define a hash member called toString (where this key is a symbol) which is a function, then that function will be called to generate a string. This is useful in many cases where hashes are used as data structures<sup>12</sup>.

# 11 Garbage collection

Garbage collection is done automatically — up to a point. Specifically, the system does reference-counted garbage collection continuously, which works well in most cases. However, it is possible to create cycles:

```
[] !A  # make a list called A
[?A] !B  # make a list called B, referencing A
?B ?A push  # add a reference to B in A
```

Now there are two objects referencing each other — a cycle. This can happen in lists, hashes and closures. Reference-counted garbage collection will never delete these. Therefore it may be necessary in programs with a complex structure to call the full garbage collector occasionally.

This is done periodically, by default every 100000 instructions or so. This interval can be changed by writing to the autogc property with a new interval:

```
1000 !autogc
```

It can also be disabled entirely by setting autogc to a negative value. A full garbage collect can be done manually by the word

```
gc
```

Incidentally, this is the same style of garbage collection used by Python.

# 12 Functional programming

Anonymous functions are defined with brackets, which will push an object representing that function (and any closure created) onto the stack. This can then be called with call (which can be abbreviated to "@") Such functions may have parameters and local variables. For example, this is a function to run a function over a range of numbers, printing the result:

With this defined, we can now use it to show the squares of those numbers:

```
(|x:| ?x ?x *) over1to10
or more simply
```

```
(dup *) over1to10
```

<sup>&</sup>lt;sup>12</sup>However, this can sometimes go horribly wrong, particularly where debugging is involved. Use with care.

#### 12.1 Recursion

Recursion is normally achieved by calling the function by name within its own definition:

```
| :factorial |x:|
| ?x 1 = if 1 else ?x ?x 1 - factorial * then;
```

This is not possible in an anonymous function. To achieve recursion in anonymous functions, use the recurse keyword to make the recursive call:

```
1 10 range (|x:| ?x 1 = if 1 else ?x ?x 1 - recurse * then) map "," intercalate.
```

will print the factorials of the first 10 natural numbers. See below for how map performs a function on each member of an iterable to produce a list, and for how intercalate builds a string out of a list by joining string representations of its members with a separator.

#### 12.1.1 A warning

Angort has a limited return stack size of 256 frames. Recursive algorithms may run out of stack. Also, Angort may not be suitable for expressing very complex recursive functions. Consider for example the quicksort algorithm: this can be implemented as

```
| :qs |1:p|
| ?1 len 1 <= if
| ?1
| else
| ?1 pop !p
| ?1 (?p <) filter qs
| [?p] +
| ?1 (?p >=) filter qs +
| then;
```

but running it on a large number of items will be very slow. Try

```
|| [] 0 100000 each {rand 200000 %,} qs
```

This generates a list of 100000 integers in the range [0, 199999] and sorts them. On my laptop it takes about 7 seconds – a long time for such a simple task. This is because the algorithm recurses deeply and widely, and each recursion constructs three lists (a new single-item list for the pivot and two lists using a filter) and pastes them together, constructing a temporary list on the way. This is very inefficient. In contrast, using the built in sort word takes only 0.25s:

```
|| [] 0 100000 each {rand 200000 %,} sort
```

This is still slow, because the internal comparison operator needs to perform typechecking on each pair of elements it compares so that integers and floats in the same list will be compared correctly. However, it is a big improvement because the libc qsort function is being used.

## 12.2 Words for dealing with functions

name	stack picture	side-effects and notes	
map	(iter func – list)	apply a function to an iterable, giving a list	
reduce	(start iter func – result)	set an internal value (the accumulator) to "start", then iterate, ap-	
		plying the function (which must take two arguments) to the ac-	
		cumulator and the iterator's value, setting the accumulator to this	
		new value before moving on.	
filter	(iter func – list)	filter an iterable with a boolean function	
zipWith	(iter iter func – list)	iterate over both iterables, combining the elements with a binary	
_		function and putting the results into a list	

We can now list all the squares of a range of numbers:

```
0 100 range (dup *) map each {i.}
```

We can also write a function to sum the values in an iterable using reduce:

#### 12.3 Closures

Anonymous functions can refer to variables in their enclosing function or function, in which case a closure is created to store the value when the enclosing function exits. This closure is mutable - its value can be changed by the anonymous function. For example, consider the following function:

This returns an anonymous function which refers to the local variable inside the function which created it. We can run this function and store the returned function in a global:

a few times, we will see an incrementing count - the value in the closure persists and is being incremented. We can call mkcounter several times and each time we will get a new closure.

#### 12.3.1 Closures are by reference

However, all closures are *by reference* — the child functions get references to variables closed in the parent function, so a function which returns several functions will all share the same closure, and any changes to variables in the closure will be reflected in all the other functions. For example:

```
| :mklistoffunctions |:x|
       []
       0 10 range each {
            i !x (?x),
       }
       ;
```

looks like it should produce a list of functions, each of which returns the numbers from 0 to 9. However, all the functions will return 9 because they all share the same copy of x. We can get around this by using a *closure factory* function to hold private copies:

```
:factory |x:| (?x);
:mklistoffunctions |:x|
[]
```

```
0 10 range each {
    i factory,
}
;
```

#### 12.3.2 Objects with delegates and private members using hashes and closures

Languages like C# have a "delegate" type, which consists of the pointer to an object and a method to be called inside that object, as a single entity.

We can emulate the same thing in Angort by creating objects as hashes, with methods as functions closing over the local variables in the creating function:

```
# create a rectangle object as a hash, with delegates to draw it
:mkrectangle |x,y,w,h:this|
    [ %
         # a delegate to draw the rectangle - this
         # will create a closure over x, y, w, h.
         'draw (?x ?y ?w ?h graphics$drawrect),
         # another to move it by some amount
         'move (|dx,dy:|
             2x \cdot 2dx + 1x
             ?y ?dy + !y
    ]
# create it
100 100 10 10 mkrectangle !R
# draw it
?R?'draw@
# move it and redraw
20 20 ?R? 'move@
?R?'draw@
```

This also provides a "private member" mechanism: if a value is defined in a closure in the function which creates the hash, rather than in the hash itself, that value will only be visible to functions defined in the hash.

# 13 Exceptions

Exception handling is done with a try/catch/endtry construction:

```
try
...
catch:symbol1,symbol2...
...
endtry
```

Upon entry to the catch block, the stack will hold the exception symbol on the top with extra data (typically a string with more information) under that. There are quite a few built-in exceptions: they are listed in <code>exceptsymbs.h.</code> To throw your own exception, use

```
throw (data except --), e.g.

| "Can't open file: " ?fn + `badfile throw
```

### 14 More built-in words

# 15 Getting help

There are many other functions and operations available in Angort. These can be listed with the list word, and help can be obtained on all words with ?? (except the very low-level words compiled directly to bytecode, which are all covered above):

```
1 \mid 0 > ?? filter filter: (iter func -- list) filter an iterable with a boolean function 1 \mid 0 >
```

If you have loaded a library or package and not imported the functions into the main namespace (see section 16.2below), you can use the fully qualified name to get help:

```
1|0 > 'io library drop
1|0 > ??io$open
io$open: (path mode -- fileobj) open a file, modes same as fopen()
1|0 >
```

# 16 Namespaces and packages

Build a package by putting the package directive at the start of a file along with the package name, e.g.

```
package wibble
```

and including that file with require instead of the usual include.

This will cause a new namespace to be created where package is invoked, and all subsequent definitions until the end of the file will be put into that namespace. On return from the file, require will put an integer handle for that namespace on the stack. All the globals, constants and functions defined in the package are available by prefixing the name with the package name and a dollar sign:

```
package$name
```

Note that the package name is that given to the package directive, not the name of the file!

### 16.1 Importing packages into the default space

With this handle on the stack, we can import the namespace — either all of it or part of it. The import word takes two forms:

```
require "wibble.ang" import
```

will import all the public names defined in the package, while

```
require "wibble.ang" ['foo, 'bar] import
```

will only import the given names — in this case, foo and bar.

The namespace handle can also be used in other ways:

name	stack picture	side-effects and notes
require "filename"	(– handle)	load a package
package packagename		directive to start a new namespace
private		all subsequent names are not exportable from the
		namespace
public		all subsequent names are exportable from the
		namespace (default)
import	(handle –)	import all public definitions from a namespace
		into the default namespace
import	(handle list –)	import some public definitions from a namespace
		into the default namespace
names	(handle – list)	get a list of names in a namespace
ispriv	(handle name – bool)	return true if a name is private in the namespace
isconst	(handle name – bool)	return true if a name is constant in the namespace

#### 16.1.1 Local packages

It is sometimes necessary to create a package inside a script which is not included from another script. One example is where the script does not terminate, leaving the user at the prompt with some functions defined, but some functions should be private.

To do this, define the package thus:

```
package somename
private
...private functions...
public
...public functions...
endpackage import
```

The endpackage word does the same as the return from a require normally does — close off the package and stack the package ID.

#### 16.2 Namespaces, libraries and modules

Native functions created by using the shared library plugin system are loaded into namespaces, just as described above. However, native functions defined using Angort's built-in function registration system are loaded into *modules*. These are very similar to namespaces, but are slightly faster<sup>13</sup>.

Such functions are automatically part of the default namespace, but can be overriden by your own functions. If you still wish to use the standard definition, the fully specified name can be used. This is in the form module\$name, like that used for libraries or plugins. For example, if you have overriden the p function, you can use std\$p.

# 17 Plugin libraries (currently Linux only)

These are shared libraries which can be loaded into Angort. They communicate with Angort via a simplified interface, and are easy to write in C++ (see the plugins directory for some examples). Once created, they should be named with the .angso extension and placed in Angort's library search path (a colon separated string). By default, this is

```
.: ~/.angort:/usr/local/share/angort
```

but it can be changed using the searchpath property (note that ~ will be expanded to the user's home directory). For example, to append /home/foo/bin to the path, you could write

 $<sup>^{13}</sup>$ The pointers to the native functions are stored directly rather than being wrapped in Value objects.

```
?searchpath ":/home/foo/bin" + !searchpath
```

Plugin libraries are loaded using the library word, which leaves a namespace handle on the stack so that import can be used, or drop to not import anything. The namespace is the library name, which may not be that of the library file — it's defined in the plugin code. For example, to load the standard file IO library and import it, we would write

```
'io library import
```

**Note that** unlike package loading with require, the library name comes first. This is because library is an instruction which is compiled and then run, rather than a compiler directive which is acted on at compile time.

This means we can do things like

```
['id3, 'mpc, 'io] each {i library import}
```

to import lists of libraries.

# 18 Topics missing from this document

There are several topics which will be discussed in later versions of this document:

- recursion
- def and defconst
- parameter type checking
- routines (generators) and yield
- constant expressions using << and >>
- sourceline
- exception and signal handling
- heredocs and barewords
- many more built-in words
- writing words and binary operators in C++ and adding them to Angort remember the gotchas: can't use "a" as a variable and must put brace on newline at start of word. Also mention wrappers.h.
- properties (which look like global variables but invoke Angort code)
- embedding an Angort interpreter in other systems

# Index

(, 23	help, <mark>27</mark>
), 23 ,, 22 ??, 27 @, 23 [, 7 [%, 22 [%], 21 \$, 27 {, 14 ], 7 ', 21, 22	i, 15 icur, 17 idone, 17 if, 12 ifirst, 17 ifleave, 14 import, 27, 28 in, 20, 22 inext, 17 isconst, 28 ispriv, 28
abs, 12 and, 11 anonymous functions, 23 arithmetic, 11 autogc, 23	<pre>ival, 21  j, 16  jval, 22</pre>
	k <b>, 16</b>
band, 11 binary operations, 11 bnot, 12 bor, 11 bxor, 11 call, 23	leave, 14 libraries, 28 library, 28 lists, 20 loop infinite, 14 iterator, 15
case, 14 cases, 14	loops, 14
closures, 25 const, 19 const functions, 19	map,24 mkiter,17
delegates, 26 drop, 11 dup, 9, 11 each, 14	names, 28 namespace, default, 28 namespaces, 27 neg, 12 not, 12
else, 12 endpackage, 28	objects, 26 or, 11
filter, 24 forward declaration, 20	otherwise, 14 over, 11
frange, 7 functional programming, 23	package, 27, 28 packages, 27
garbage collection, 23 gc, 23 get, 20, 22 global, 18	local, 28 pop, 20 private, 28 private member, 26 public, 28
hashes, 21 shortcut set/get, 22	push, 20 range, 7

```
reduce, 24
remove, 20, 22
require, 27, 28
searchpath, 28
set, 20, 22
shift, 20
stack, 6
manipulation, 11
swap, 11
symbols, 21
then, 12
toString, 23
unshift, 20
word, 7
zipWith, 24
```